

A word about spreadsheets – what are they?

In 1979, when the first edition of *Practical Astronomy with your Calculator* was published, very few people had access to a computer. Although home computers were beginning to appear in the high street, they were not the commonplace household accessory we see today. Calculations were made using a calculator, the sophistication of which ranged from the simple four-function device to the versatile programmable reverse-Polish scientific machine. You may already own a calculator that would be suitable for the recipes given here, but you might also own a computer and wish to make the calculations using that instead. If you are good at programming, you could consider using the methods described in this book as a basis for writing your own astronomical software. But most of us don't want to embark on such a project. How then can we use our computers to make astronomical calculations?

One answer is to use a **spreadsheet** program such as Microsoft's *Excel*, or OpenOffice *Calc*. The latter is available at no cost, and described as fully compatible with the former, so if you do not already own a commercial spreadsheet program, then *Calc* might be a good way to go. Once you have loaded the software on to your machine, open the spreadsheet program. The screen display should then look something like Figure I. (Here and throughout the book, toolbars, sidebars and many other features have been removed from the spreadsheet views.)




Figure I. An empty spreadsheet.





Figure II. Cell C5 carries the number 23.9, and cell D5 carries the label This is a number.

The spreadsheet consists of an array of cells, labelled A, B, C etc. across the top (these are the column labels) and 1, 2, 3 etc. down the left-hand side (these are the row labels). Each individual cell is labelled by its column letter and its row number, e.g. A1, B25 etc. The cell with the thick border around it in Figure I is cell C5. You can write some text or numbers in any cell. In Figure II, the number 23.9 has been placed in cell C5, and the label This is a number has been placed in cell D5. (Since cell E5 is empty, the program has allowed the label to overwrite the space allocated to E5, although the entire content This is a number remains in D5, and E5 remains empty.) The spreadsheet knows that something placed in a cell is a label (i.e. text) if you begin the entry with a single apostrophe symbol ('). If you want to enter a number as a number, just type it in. If you want the spreadsheet to treat the number as a label, put the apostrophe in front of it.


We can obviously put labels and numbers in any of the cells, but the real power of the spreadsheet comes from using formulas. A formula is a calculation which can use the contents of other cells. The result of the calculation is displayed in the cell carrying the formula, so you are not usually aware of the calculation that has gone on in the background since what is displayed is the result rather than the formula itself. A formula is placed in a cell by typing the equals sign (=) followed by the formula. The spreadsheet knows from the equals sign that it is to calculate the formula and display the result. For example, in Figure III, cell C6 carries the entry $=C5 * C5$. You will see that C6 now displays the result of multiplying the number in cell C5 by itself (the star symbol * means 'multiply'), i.e. the square of the number 23.9, which is the number 571.21. We have also placed the label This is its square in cell D6.



	A	B	C	D	E	F	G	H	I	J	K	L
1												
2												
3												
4												
5			23.9	This is a number								
6			571.21	This is its square								
7												
8												
9												
10												
11												
12												
13												
14												
15												

Figure III. Cell C6 carries the formula $=C5*C5$ and hence displays the square of 23.9.

Let's see what happens if now we change the number in cell C5 without making any other change to the spreadsheet. In Figure IV the number in C5 has been changed to the number 4.0 and, hey presto, the square of 4 (i.e. 16) is displayed in cell C6. You can begin to see that complex calculations can be performed for you automatically with a spreadsheet program. With the right formulas placed in order in the spreadsheet, the results can be calculated for any set of starting values. That is just what we want to do in this book. We can hide the complications of the calculation of, say, the time of sunrise within the formulas and just enter a date and geographical location in the correct cells at the top to obtain the result immediately.



	A	B	C	D	E	F	G	H	I	J	K	L
1												
2												
3												
4												
5			4	This is a number								
6			16	This is its square								
7												
8												
9												
10												
11												
12												
13												
14												
15												

Figure IV. Cell C5 now carries the number 4 and so cell C6 displays the number 4 multiplied by 4 which is 16.

We don't need to explain much more about spreadsheets here, although we will note various techniques as we go along. If you want to learn more about their powerful capabilities we suggest buying a book about spreadsheets (see the Bibliography on page 208 for a suggestion). In this book, we have supplied you with the spreadsheet and formulas for most calculations, so all you have to do is to type in the labels, numbers and formulas as shown. The spreadsheet will then do its work automatically and give you the answer for

any starting values you enter. (We have provided the spreadsheets ready-made on our website. Please look in the section “A useful website” on page 209 for details.)

The layout of spreadsheets in this book

All of the spreadsheets in this book conform to the same general format (see Figure V). At the top, in cell A1, is the title of the spreadsheet (in this case Converting decimal hours to hours, minutes and seconds). It is best to use a slightly larger font size for this and to make it boldface as here. We have used Arial 16 point for the title. Row number 2 is left blank (i.e. none of the cells has anything in it). In row 3, we have written the label *Input* in A3 (Times New Roman font, italic face, 10 point) to remind us that the input values for the spreadsheet are entered to the right of this cell. In the case shown in Figure V, there is only one input value, the decimal hours (name label in B3, Arial font, bold face, 14 point), and it is entered in cell C3 (also Arial font, bold face, 14 point). In spreadsheets which have more than one input value, the others have their name labels in cells B4, B5 etc. and their corresponding values in C4, C5 etc.

	A	B	C	D	E	F	G	H	I
1	Converting decimal hours to hours, minutes and seconds								
2									
3	<i>Input</i>	decimal hours	18.52416667			<i>Output</i>	hours	18	=C14
4							minutes	31	=C12
5							seconds	27	=C10
6	<i>Spreadsheet label</i>								
7	1	unsigned decimal	18.52416667	=ABS(C3)					
8	2	total seconds	66687	=C7*3600					
9	3	seconds (2 dp)	27	=ROUND(MOD(C8,60),2)					
10	4	corrected seconds	27	=IF(C9=60,0,C9)					
11	5	corrected remainder	66687	=IF(C9=60,C8+60,C8)					
12	6	minutes	31	=MOD(TRUNC(C11/60),60)					
13	7	unsigned hours	18	=TRUNC(C11/3600)					
14	8	signed hours	18	=IF(C3<0,-1*C13,C13)					

Figure V. The layout of a spreadsheet.

The results of the calculations, i.e. the output values, are provided to the right of cell F3. We have written the label *Output* in F3 (Times New Roman font, italic face, 10 point) to remind us that the output values calculated by the spreadsheet appear to the right of this cell. In the case shown in Figure V there are three output values, called hours, minutes, seconds. Their name labels appear in cells G3, G4, G5 (Arial font, bold face, 14 point) and their values in H3, H4, H5 (also Arial font, bold face, 14 point) respectively. Just to the right of the three output values, in column I, are shown the formulas (written as labels, i.e. with an

apostrophe in front of the equals sign to stop the program calculating the formula) that are actually in the output value cells. Thus cell H3 actually contains the formula =C14 (i.e. it will display the value of the cell C14) and you will need to enter =C14 in the cell H3. Wherever you see a formula (anything beginning with the equals sign) enter exactly that formula in the cell immediately to its left. In this case you would put =C14 in cell H3, =C12 in cell H4, and =C10 in cell H5.

The calculations carried out by the spreadsheet begin on row 7 in Figure V. Each row corresponds to one step in the calculation, in this case the calculation method of Section 8. In the method table shown in that section there are just two steps, whereas in the spreadsheet there are eight. There is only a rough correspondence between method steps and spreadsheet steps. This is partly because the spreadsheet calculations do not have the benefit of human intelligence to assist them! For example, if you used your calculator to carry out the steps of Section 8, and you found that the result was, say, 6h 35m 60s, you would automatically write this as 6h 36m 0s. The spreadsheet would, however, quite happily report the result in the first format. We get over the problem in the spreadsheet by first stripping out the sign, then converting to seconds, then finding the seconds, minutes and hours in that order, and finally putting back the sign.

In the example shown in Figure V, you would enter the labels and formulas exactly as shown. Thus on row 7 you place the label '1 in A7 (this is text, and the apostrophe tells the spreadsheet so), the label 'unsigned decimal in B7 and the formula in C7 shown immediately to its left, i.e. =ABS(C3). Do this for each calculation row (7 to 14 in this case). Finally, rename the spreadsheet on the tab at the bottom (DHHMS in this case). (You can probably do this by pointing at it with the mouse, pressing the right-hand mouse button, and selecting the 'rename' option.)

Although the labels in columns A and B make no difference to the calculations, we recommend that you put them in as they make the spreadsheet much easier to understand. This becomes more important if you return to a spreadsheet some time after you constructed it.

One other note about spreadsheets that you might find useful concerns column widths. If the column width is too narrow to display the content of a cell, you may just see something like ##### displayed instead. You can adjust the column width by placing the mouse pointer on the division between the label (A, B, C etc.) of the column you want to alter and the label of the column immediately to its right, holding down the left-hand mouse button, and 'dragging' the column width left or right as needed.

Calculations involving multiple sheets

Some of the spreadsheet calculations, as in the example just given, use just one sheet. Most, however, use several. For example, suppose that a first spreadsheet calculation results in a number expressed in decimal hours but the answer has to be in the form hours, minutes and seconds. The first sheet passes its answer (in decimal hours) to a second sheet which carries out the conversion and passes the converted result back again to the first sheet.

A concrete example is illustrated by a spreadsheet for Section 14, reproduced in Figure VI. You will see that there are three tabs in the bottom left-hand corner, corresponding to three sheets labelled GSTLST, HMSDH and DHHMS. Only the top sheet, GSTLST is visible in the figure with the other two lying 'underneath' it. The input values to the calculation include the Greenwich sidereal time (GST) expressed in hours, minutes and seconds (cells C3, C4 and C5). These must first be converted to the GST expressed in decimal hours, a calculation covered in Section 7. The spreadsheet for that section, labelled HMSDH, must

	A	B	C	D	E	F	G	H	I
1	Conversion of GST to LST								
2									
3	<i>Input</i>	GST hours	4		<i>Output</i>	LST hour	0	=C12	
4		GST mins	40			LST min	24	=C13	
5		GST secs	5.23			LST sec	5.23	=C14	
6		geog long	-64						
7									
8	1	GST	4.668119444	=HMSDH!H3					
9	2	offset	-4.266666667	=C6/15					
10	3	LST (hours)	0.401452778	=C8+C9					
11	4	LST (hours)	0.401452778	=C10-(24*INT(C10/24))					
12	5	LST hour	0	=DHHMS!H3					
13	6	LST min	24	=DHHMS!H4					
14	7	LST sec	5.23	=DHHMS!H5					

Figure VI. A spreadsheet with multiple sheets.

	A	B	C	D	E	F	G	H	I
1	Converting hours, minutes and seconds to decimal hours								
2									
3	<i>Input</i>	hours	4	=GSTLST!C3	<i>Output</i>	decimal hours	4.668119444	=C10	
4		minutes	40	=GSTLST!C4					
5		seconds	5.23	=GSTLST!C5					
6									
7	1	A	0.087166667	=ABS(C5)/60					
8	2	B	0.668119444	=(ABS(C4)+C7)/60					
9	3	C	4.668119444	=ABS(C3)+C8					
10	4	D	4.668119444	=IF((C3<0)+(C4<0)+(C5<0),-C9,C9)					

Figure VII. Illustrating cross-references between sheets.

be included in this spreadsheet file as an additional sheet – the tab HMSDH in Figure VI. Figure VII shows the spreadsheet with the HMSDH sheet on top so it is visible.

The link between the sheets is accomplished by using the sheet name, followed by an exclamation mark (!) and the cell reference. In Figure VII, the input value of hours (C3) is obtained from cell C3 of sheet GSTLST by using the formula =GSTLST!C3. Similarly, the input value of minutes is obtained using the formula =GSTLST!C4 in cell C4 of HMSDH, and the input value of seconds is obtained by using the formula =GSTLST!C5 in cell C5. The result of the calculation by this sheet, the decimal hours, appears in cell H3

of Figure VII. This is passed back to sheet GSTLST in cell C8 of Figure VI, which contains the formula =HMSDHIH3.

Similarly, the result of the calculation of GSTLST, expressed in decimal hours, appears in cell C11 of Figure VI. This needs to be converted to the format hours, minutes and seconds and it is passed to sheet DHHMS (see Figure VIII) by using the formula =GSTLST!C11 in cell C3 of that sheet. Sheet GSTLST then extracts the results from sheet DHHMS (H3, H4 and H5 of Figure VIII) using the formulas DHHMS!H3, DHHMS!H4 and DHHMS!H5 respectively in cells C12, C13, and C14 of Figure VI.

File Edit View Insert Format Tools Data Window Help									
Type a question for help									
C3	A	B	C	D	E	F	G	H	I
1	Converting decimal hours to hours, minutes and seconds								
2									
3	Input	decimal hours	0.401452778	=GSTLST!C11		Output	hours	0	=C14
4							minutes	24	=C12
5							seconds	5.23	=C10
6									
7	1	unsigned decimal	0.401452778	=ABS(C3)					
8	2	total seconds	1445.23	=C7*3600					
9	3	seconds (2 dp)	5.23	=ROUND(MOD(C8,60),2)					
10	4	corrected seconds	5.23	=IF(C9=60,0,C9)					
11	5	corrected remainder	1445.23	=IF(C9=60,C8+60,C8)					
12	6	minutes	24	=MOD(TRUNC(C11/60),60)					
13	7	unsigned hours	0	=TRUNC(C11/3600)					
14	8	signed hours	0	=IF(C3<0,-1*C13,C13)					

Figure VIII. Illustrating cross-referencing between sheets.

Now you can proceed in this way if you wish, using multiple sheets to carry out specific calculations as just described, but the result can be quite confusing when you have a complicated calculation requiring many sheets. A better way to proceed is for us to define our own functions and use these instead to carry out the calculations. This is the approach that we have adopted here.

Using our own functions

Microsoft *Excel* and OpenOffice *Calc* both come with an internal programming language called BASIC. We don't need to go in to any of the details of what this is and how it works, but suffice it to say that we have written functions to carry out most of the calculations described in this book. All you have to do is to use the functions in your spreadsheet exactly as if they were formulas. This has the advantage that you now need only one sheet for any calculation with no cross-linking to multiple sheets, making the whole thing easier to comprehend. Another advantage is that we have provided functions with much higher accuracy than the simplified calculations of many of the sections. For example, you can use the method of Section 46 to calculate the Sun's ecliptic longitude approximately, or you can use the function SunLong to calculate it much more precisely.

Let us illustrate the use of functions instead of multiple sheets using the example above. Figure IX shows the spreadsheet of Section 14 using functions instead of multiple sheets. Compare this with Figure VI. You can see that in Figure IX there is now only one sheet, labelled GSTLST.

	A	B	C	D	E	F	G	H	I
1	Conversion of GST to LST								
2									
3	<i>Input</i>	GST hours	4		<i>Output</i>	LST hour	0	=C12	
4		GST mins	40			LST min	24	=C13	
5		GST secs	5.23			LST sec	5.23	=C14	
6		geog long	-64						
7									
8	1	GST	4.668119444	=HMSDH(C3,C4,C5)					
9	2	offset	-4.266666667	=C6/15					
10	3	LST (hours)	0.401452778	=C8+C9					
11	4	LST (hours)	0.401452778	=C10-(24*INT(C10/24))					
12	5	LST hour	0	=DHHour(C11)					
13	6	LST min	24	=DHMin(C11)					
14	7	LST sec	5.23	=DHSec(C11)					

Figure IX. Illustrating the use of functions instead of multiple sheets.

The results of the calculation, contained in cells H3, H4 and H5 in both Figures VI and IX, are identical, but in place of the cross-references between sheets at C8, C12, C13 and C14 of Figure VI there are formulas in the corresponding cells of Figure IX. In cell C8, for example, the formula =HMSDH(C3,C4,C5) converts the hours, minutes and seconds (in cells C3, C4 and C5) to decimal hours, with the result shown in cell C8. The contents of C3, C4 and C5 are passed to the function HMSDH as the references contained within the brackets after the function. When the spreadsheet program sees a formula, in this case =HMSDH(C3,C4,C5), it first looks through a list of its own formulas, and then checks to see if the function has been written in BASIC. If it has, the spreadsheet then runs the BASIC program corresponding to the function, passing the contents of the cells in the reference list to the BASIC program, in this case the contents of cells C3, C4 and C5. The result of the calculation is then passed back to the spreadsheet where it appears in the same cell as the function (C8).

Functions like this have been provided for most of the calculations in this book, and are described in the corresponding sections. You will need to download the spreadsheets from the Cambridge University Press website in order to obtain the functions (which are included invisibly with each sheet). Please look in the section “A useful website” on page 209 for details.